微机在物理实验中的应用

蔡美娟　潘国品

封面设计：朱锺民

研究与探索

（微机在物理实验中的应用）专辑　（内部发行）

编辑者：《研究与探索》编辑部　　印 刷 者：武进县村前印刷厂

出版者：上 海 交 通 大 学　　发行和订购：上海交通大学科技交流室

成本费 1.40 元

前　言

当今，我们处在信息爆炸和新技术革命迅猛发展的时代，微型计算机在各方面的应用显示了它特有的优越性，微机的应用正在日趋普及，微机在物理实验中的利用也引起了人们的注意。近来，我们在应用微机进行物理实验数据处理方面做了一些探索和实践，感到应用微机来进行辅助教学更有其重要意义。在物理实验中，运用微机这个先进工具进行数据处理，有利于提高学生在程序的编写和利用计算机解决实际问题的能力，为在后继课程和科学实验中使用微机创造条件；有利于老师和学生当场检验实验的结果，便于老师掌握学生的实验情况，及时发现和解决实验过程中存在的问题，提高实验质量；在某种程度上也可节省学生在数据计算，制表和作图的时间，明显地提高了效率。

本文列入了十八个物理实验的数据处理程序，程序说明和应用的例子，这些程序包括力学、电磁学和光学三个方面。除此，还编入了应用广泛的八个子程序，它们不但可作为子程序调用，也可单独使用，在许多实验中都要用到它们。

以上这些程序都是以BASICⅡ语言编写成的。BASICⅡ是微机中应用最普遍，也是最容易掌握的语言，几乎所有微机都配备了这种语言，因此本文具有广泛的适应性。这些程序都已在MIC-80微机(相当于TRS-80微机)通过计算，并且经过一届学生的教学实践。实践表明这些程序是合理的。这些程序虽然是按MIC-80微机系统编制的，但是对于其它微机只要注意它们在操作上的微小差异，稍作些改动便可以使用。对于TRS-80兼容机如：PS-85，SEED-Z80，MDR-Z80，YEE8100，BC3-80，DJS-043等型号微机，不作改动就可使用。

本文可作为工科院校师生普通物理实验的辅助资料，也可供从事数据处理工作者参考，同时对学习BASICⅡ语言和掌握编程序的技巧也有一定的帮助。

由于是初次尝试，编写的时间仓促，水平有限，难免有不妥和错误之处，希望使用者批评指正。

目　　录

第一部分　常 用 子 程 序

一、单变量统计计算

对于等精度的多次独立测量，一般以算术平均值作为待测量的最可信赖值，用$\overline{x}$表示

$$\overline{x}=\frac{\sum x_i}{n} \tag{1-1-1}$$

式中n是多次测量的次数，x_i为待测量的测量值($i=1,2\cdots\cdots n$)。为了衡量测量值的精密程度，可采用测量值的根均方差σ_x表示：

$$\sigma_x=\sqrt{\frac{1}{N-1}\sum_{i=1}^{n}(x_i-\overline{x})^2} \tag{1-1-2}$$

σ_x又称为测量值的标准误差，它还可以化为下列形式

$$\sigma_x=\sqrt{\frac{1}{N-1}[\sum x_i^2-(\sum x_i)^2/N]} \tag{1-1-3}$$

统计计算告诉我们，对于测量值中任意一个测量值x_i，在($\overline{x}+w_n\cdot\sigma_x$，$\overline{x}-w_n\cdot\sigma_x$)范围内出现的机会为99.7%，其中$W_n$是肖维勒剔除粗差的标准。

W_n可用下面经验公式表示

$$W_n=\frac{N}{(1.02+0.4\times N)} \tag{1-1-4}$$

若测量值x_i满足$|x_i-\overline{x}|>W_n\cdot\sigma_x$即为粗差，应该在测量值中剔除$x_i$值。

程序清单：

```
10 REM THE STATICS CALCULATION OF SINGLE VARIATION
20 DATA 3.42,3.43,3.44,3.94,3.44,3.43
30 INPUT "N=";N
40 DIM X(N)
50 FOR I=1 TO N
60 READ X(I): NEXT
70 N1=N: GOSUB 300
80 WN=N/(1.02+0.4*N): DW=DX*WN
90 K=0: FOR I=1 TO N
100 IF ABS(X(I)-X)>DW THEN K=K+1: X(I)=0
110 NEXT
120 IF K>0 THEN N1=N-K: GOSUB 300
130 E=DX/X*100: E=INT(E*10+0.5)/10
140 LPRINT "X="X;"(CM)","DX="DX;"(CM)","E="E;"%"
150 END
300 S=0:W=0
310 FOR I=1 TO N
320 S=S+X(I): W=W+X(I)*X(I):NEXT
330 X=S/N1:DX=SQR(ABS((W-S*S/N1)/(N1-1)))
335 DX=INT(DX*1000+0.5)/1000
340 RETURN
```

实例：

将某一物体，用一根毫米尺测量6次，测得长度的值分别为3.42，3.43，3.44，3.94，3.44，3.43厘米，求其物体的长度。

把测量值从20句DATA中输入，通过计算可知，因为3.94是满足$|x_i-\overline{x}|>W_n\cdot\sigma_x$的，即为粗差而被剔除，得到计算结果：

```
X= 3.432 (CM)    DX= 8E-03 (CM)   E= .2 %
```

二、一元线性回归分析

在科学实验中，经常需要讨论同一物理过程中的两个或多个变量之间的相互关系和变化规律。在实际问题中，或是由于这些变量之间的关系比较复杂，或是由于不可避免地存在着实验误差。变量之间的关系一般具有某种不确定性，即无法给出完全确定性的函数关系式。变量之间的这种关系称为相关关系，研究相关关系需要采用数理统计方法。回归分析就是一种用数理统计方法，从大量的实验数据中寻求同一物理过程中各变量之间的(回归)函数关系式的有效方法。

回归分析就是从一组实验数据出发，确定变量间的定量关系式，并对有关参数进行估计。实验曲线的拟合和经验公式的建立通常采用回归分析方法。

在实验中，很多自变量的数值是可以选定的非随机变量，而因变量是与自变量相互独立的随机变量，并且因变量对于自变量存在线性关系。如在伏安法测电阻的实验中，电压与电流的关系存在线性关系：V＝RI。电压，电流两个变量之间的相关关系就可用线性回归分析来处理。

这样，仅仅处理两个变量之间的关系又称一元线性回归分析。一元线性回归分析不但简单，而且应用也很广。

一元线性回归方程　　$$\hat{Y}=A_1+B_1x \qquad (1\text{-}2\text{-}1)$$

式中　$\hat{Y}$表示变量Y的估计值，又称回归值。

A_1，B_1为待定系数，又称回归系数。

(1-2-1)式是直线方程，其中A_1为截距、B_1为斜率。

设实验测量N次，X和Y的两个变量的测量值各为(x_1,y_1)，(x_2,y_2)，……，(x_n,y_n)

则有回归方程　$$\hat{Y}_i=A_1+B_1x_i \qquad (i=1,2,\cdots,N) \qquad (1\text{-}2\text{-}2)$$

令　$$Y_i-\hat{Y}=V_i=Y_i-A_1-B_1x_i \qquad (1\text{-}2\text{-}3)$$

式中，V_i代表剩差(或残差，或离差)

若$\hat{Y}_i$与Y_i的偏差越小，则回归值与测量值拟合得越好。全部测量值Y_i与回归值$\hat{Y}_i$的剩差平方和(即$\sum_{i=1}^{N}V_i^2$)刻划了偏离程度。回归方程(1-2-2)式中的待定系数A_1、B_1可由剩差平方和为最小值时求得　即

$$\sum_{i=1}^{N} V_i^2 = \min \qquad (1\text{-}2\text{-}4)$$

(1-2-4)式又可写成：

$$Q_{剩} = \sum_{i=1}^{N}(Y_i - A_1 - B_1x_i)^2 = \min \qquad (1\text{-}2\text{-}5)$$

(1-2-5)称高斯原则，由(1-2-5)式确定回归系数A_1，B_1的方法称最小二乘法。

用最小二乘法配定的直线与用其它方法配定的直线相比，它与所有N个测量值的偏离程度是最小的。

根据多元函数极值的必要条件

A_1与B_1应满足如下公式

$$\begin{cases} \dfrac{\partial Q_{剩}}{\partial A_1} = -2\sum\limits_{i=1}^{N}(Y_i - \widehat{Y}_i) = -2\sum\limits_{i=1}^{N}(Y_i - A_1 - B_1x_i) = 0 \\ \dfrac{\partial Q_{剩}}{\partial B_1} = -2\sum\limits_{i=1}^{N}(Y_i - \widehat{Y}_i)x_i = -2\sum\limits_{i=1}^{N}(Y_i - A_1 - Bx_i)x_i = 0 \end{cases} \qquad (1\text{-}2\text{-}6)$$

再把(1-2-6)式化简得：

$$\begin{cases} NA_1 + B_1\sum\limits_{i=1}^{N}x_i = \sum\limits_{i=1}^{N}Y_i \\ \sum\limits_{i=1}^{N}x_iA_1 + B_1\sum\limits_{i=1}^{N}x_i^2 = \sum\limits_{i=1}^{N}x_iY_i \end{cases} \qquad (1\text{-}2\text{-}7)$$

则此(1-2-7)式称为正规方程

令： $u = B_1$ $\qquad v = A_1$

$A = \sum\limits_{i=1}^{N}x_i^2$ $\qquad B = \sum\limits_{i=1}^{N}x_i$

$C = \sum\limits_{i=1}^{N}x_i \cdot y_i$ $\qquad D = \sum\limits_{i=1}^{N}y$

(1-2-7)式正规方程又可写成如下形式：

$$\begin{cases} Au + Bv = C \\ Bu + Nv = D \end{cases} \qquad (1\text{-}2\text{-}8)$$

解方程(1-2-8)式　求得u，v

将u和v代入(1-2-1)式得：

$$\widehat{Y} = ux + v \qquad (1\text{-}2\text{-}9)$$

(1-2-9)是回归直线方程，其中u是直线的斜率，v是直线的截距。

程序清单：

```
10 REM LINEAR REGRESSION OF SINGLE VARIABLE
20 INPUT "N=";N
30 FOR I=1 TO N
35 READ X,Y
40 A=A+X*X: B=B+X
50 C=C+X*Y: D=D+Y
60 NEXT
70 G=N*A-B*B
80 IF G=0 PRINT"NO UNIQUE SOLUTION": END
90 U=(N*C-B*D)/G: V=(A*D-B*C)/G
100 PRINT"Y="U;"*X+";V
110 LPRINT"THE REGRESSION EQUATION IS :"
120 LPRINT "Y="U;"*X+";V
130 LPRINT " ": LPRINT "R="INT(U*100+0.5)/100
140 DATA .01015,1,.02025,2,.0305,3,.0405,4,.05125,5,.061,6,.07125,7
```

实例：

在“伏安法测定电阻”的实验中，实验测量的电压和电流的数据如下

次　数	1	2	3	4	5	6	7
I (A)	0.01015	0.02025	0.0305	0.0405	0.05125	0.0610	0.07125
u (V)	1.000	2.000	3.000	4.000	5.000	6.000	7.000

将这些值用140句DATA语句输入计算机

计算的结果：

```
THE REGRESSION EQUATION IS :
Y= 98.0503 *X+ 9.35326E-03

R= 98.05
```

三、一元非线性回归分析

在实际问题中，两个变量之间的回归关系大多数是非线性的。但是，在许多情况下，非线性回归可以通过简单的变量变换，转化为线性回归模型来解。

例 〈1〉热敏电阻与温度的关系

公式　$R_t = ae^{\frac{b}{T}}$　(1-3-1)

把(1-3-1)式两边取对数　$\lg R_t = \lg a + \frac{b}{T}$　(1-3-2)

令　$Y = \lg Rt$；$A_1 = \lg a$；$B_1 = b$　　$X = \frac{1}{T}$

则(1-3-1)式成为　$Y = A_1 + B_1 X$　(1-3-3)

(1-3-3)是直线方程，它就可以利用一元线性回归分析来介。

〈2〉气体的压强和体积之关系

由波意耳定律可知：$P \cdot V = b$(常数)　　(1-3-4)

(1-3-4)式是一根双曲线。

令　$Y = P$；　$X = \frac{1}{V}$；　$B_1 = b$；　$A_1 = o$

则(1-3-4)式成为：　$Y = B_1 X$　　(1-3=5)

(1-3-5)是直线方程，也可采用一元线性回归分析来解。

〈3〉单摆中摆长与周期的关系

公式　$L = \frac{g}{4\pi^2} T^2$　　(1-3-6)

令　$Y = L$；　$X = T^2$；　$B_1 = \frac{g}{4\pi^2}$；　$A_1 = o$

则(1-3-6)式成为　$Y = B_1 X$　　(1-3-7)

(1-3-7)是直线方程，也可采用一元线性回归分析来解。

〈4〉物体自然对流冷却中温度与时间的关系

冷却定律　$\frac{\Delta\theta}{\Delta t} = \frac{K}{mc}(\theta - \theta_0)^\alpha$　　(1-3-8)

把(1-3-8)式两边取对数

$$\lg \frac{\Delta\theta}{\Delta t} = \alpha \lg(\theta - \theta_0) + \lg \frac{K}{mc} \qquad (1\text{-}3\text{-}9)$$

令　$Y = \lg\frac{\Delta\theta}{\Delta t}$，　$X = \lg(\theta - \theta_0)$；　$B_1 = \alpha$；　$A_1 = \lg\frac{K}{mc}$

则成为　$Y = A_1 + B_1 X$　　(1-3-10)

(1-3-10)是直线方程，也可采用线性回归分析来解。

〈5〉电容器放电过程中电量与时间的关系

公式　$q = Qe^{-\frac{t}{RC}}$　　(1-3-11)

把(1-3-11)式两边取对数

$$\ln q = -\frac{1}{RC} t + \ln Q$$

令　$Y = \ln q$；　$B_1 = -\frac{1}{RC}$；　$X = t$；　$A_1 = \ln Q$

则成为　$Y = B_1 X + A_1$　　(1-3-12)

(1-3-12)是直线方程，也可采用一元线性回归分析来解。

实例：

单摆实验中摆长与周期的关系：

$$L = \frac{g}{4\pi^2} T^2 \qquad (1\text{-}3\text{-}13)$$

式中：　L 为摆长　　T 为周期　　g 是需求的重力加速度

(1-3-13)式表示L与T关系是一条抛物线

只要作如下变换：　　令　　$T_1 = T^2$,　　$\frac{g}{4\pi^2} = u$

　　则　　　　$L = u \cdot T_1$　　就有直线关系了

可以采用一元线性回归分析求得 u

　　则　　　　$g = 4\pi^2 \cdot u$

已知测量数据如下表：

X(I)　(cm)		105.1	115.0	125.0	135.0	141.0
50T (s)	1	100.33	104.91	109.76	114.34	117.23
	2	100.36	104.83	109.81	114.41	117.32
	3	100.34	104.87	109.82	114.25	117.22

$$X_0 = 4.20\text{cm} \qquad d_0 = 2.194\text{cm}$$

$$L = X(I) - X_0 - \frac{d_0}{2}$$

$$T = \sum_{i=1}^{3} (50T)_i / 3/50$$

$$N = 5, \qquad M = 3, \qquad K = 50$$

程序清单：

```
10 REM NOLINEAR REGRESSION OF SINGLE VARIABLE
20 DATA 105.1,115,125,135,141
30 DATA 100.33,100.36,100.34,104.91,104.83,104.87,109.76,109.81
35 DATA 109.82,114.34,114.41,114.25,117.23,117.32,117.22
50 INPUT "N,M,K,X0,D0=";N,M,K,X0,D0
55 DIM X(N),T(N,M),L(N),T1(N),D(N)
60 FOR I=1 TO N
70 READ X(I): L(I)=X(I)-X0-D0/2
80 NEXT
90 FOR I=1 TO N
100 FOR J=1 TO M
110 READ T(I,J)
120 NEXT J,I
130 FOR I=1 TO N: S=0: FOR J=1 TO M
140 S=S+T(I,J): NEXT J
150 T1(I)=S/K/M: D(I)=T1(I)*T1(I): NEXT I
160 FOR I=1 TO N
170 A=A+D(I)*D(I): B=B+D(I): C=C+D(I)*L(I): D1=D1+L(I)
180 NEXT
190 G1= (N+1)*A-B*B
200 U=((N+1)*C-B*D1)/G1: V=(A*D1-B*C)/G1
210 G=4*3.14159*3.14159*U: G=INT(G*10+0.5)/10
220 PRINT"G="G
230 LPRINT "G="G;"CM/S↑2"
```

计算结果：

```
G= 977.6 CM/S↑2
```

四、二变量统计计算

若对 x，y 两变量独立测量 n 次，则可得到 (x_1,y_1)，$(x_2,y_2)\cdots\cdots(x_n,y_n)$。如果认为 x，y 间有线性关系，则可写成线性回归方程，$y=A_1+B_1x$ 。由最小二乘法可证明各 (x_i, y_i) 值偏离 $y=A_1+B_1x$ 的大小可用剩余标准误差 σ_s 表示：

$$\sigma_s=\sqrt{\frac{(1-R^2)S_{YY}}{n-2}} \qquad (1-4-1)$$

其中R称为相关系数可表示为：

$$R=\frac{S_{xy}}{\sqrt{S_{xx}S_{YY}}} \qquad (1-4-2)$$

而 S_{xx}，S_{YY}，S_{xy} 分别为：

$$S_{xx}=\sum x^2-\frac{(\sum x)^2}{n} \qquad (1-4-3)$$

$$S_{YY}=\sum y^2-\frac{(\sum y)^2}{n} \qquad (1-4-4)$$

$$S_{XY}=\sum xy-\frac{\sum x\sum y}{n} \qquad (1-4-5)$$

用统计方法还可以求得截距A_1和斜率B_1以及它们的标准误差σ_A和σ_B，其中σ_A,σ_B分别为

$$\sigma_B=\frac{\sigma_S}{\sqrt{S_{xx}}} \qquad (1-4-6)$$

$$\sigma_A=\sigma_B\cdot\sqrt{\frac{\sum x^2}{n}} \qquad (1-4-7)$$

在二个变量统计计算中 σ_s 的数值描绘了线性回归方程的精密度，可以证明测量点(x_i, y_i)落在 $y=(A_1+B_1x)\pm W_n\cdot\sigma_s$ 范围内的机率是 99.7%，若测量值 (x_i, y_i) 满足 $|Y_i-(A_1+B_1x_i)|>W_n\cdot\sigma_s$ 即为粗差，应该在测量列中剔除它，其中 W_n 是肖维勒剔除粗差标准，W_n可用经验公式表示

$$W_n=\frac{n}{(1.02+0.4\times n)}$$

式中 n 为测量次数

一般情况下，若能估计粗差不存在时，可以不考虑剔除粗差问题，程序就简单了。

实例：

研究钢丝受拉力而伸长的长度ΔL与所加的拉力 F之关系，ΔL 用镜尺法测量，

$$l=\frac{2D}{K}\cdot\Delta L \qquad\qquad F=C\cdot l$$

已知测量值为F_i与 l_i，求其 F 与 l 的比例系数 C

数据表：

F (kg)	0	0.5	1.0	1.5	2.0	2.5	3.0
l (cm)	0	0.39	0.79	1.18	1.58	1.98	2.38

程序清单：

```
5 REM THE STATICS CALCULATION OF DOUBLE VARIATION
10 DATA 0,0,0.5,0.39,1.0,.79,1.5,1.18,2.0,1.58,2.5,1.98,3,2.38
20 INPUT "N=";N
25 FOR I=0 TO N
30 READ F,L
40 A=A+L*L:B=B+L:C1=C1+L*F
50 D=D+F:E=E+F*F:NEXT
60 G=N*A-B*B
70 U=(N*C1-B*D)/G:V=(A*D-B*C1)/G
80 S1=A-B*B/N:S2=E-D*D/N:S3=C1-B*D/N
90 R=S3/SQR(S1*S2)
100 DS=SQR((1-R*R)*S2/(N-2))
110 DC=DS/SQR(S1):DC=INT(DC*1000+0.5)/1000
120 E=DC/U*100:C=INT(U*1000+0.5)/1000
125 IF E)1 THEN E=INT(E+0.5)
126 IF E)1E-2 THEN E=INT(E*100+0.5)/100
127 E=INT(E*1E4+0.5)/1E4
130 PRINT"C="C;"+-";DC,"E="E;"%"
140 LPRINT "C="C;"+-";DC,"E="E;"%"
```

计算结果：

```
C= 1.257 +- 2E-03                E= .16 %
```

五、用高斯法解线性方程组

设有线性代数方程组为

$$\begin{cases} a_{11}x_1+a_{12}x_2+\cdots\cdots\cdots\cdots+a_{1n}x_n=b_1 \\ a_{21}x_1+a_{22}\mathrm{n}_2+\cdots\cdots\cdots\cdots\cdots+a_{2n}x_n=b_2 \\ \cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots \\ a_{n1}x_1+a_{n2}x_2+\cdots\cdots\cdots\cdots\cdots\cdots+a_{nn}x_n=b_n \end{cases} \tag{1-5-1}$$

它可以简写成 $$\sum_{j=1}^{n}A_{i,j}x_j=b_i(i=1\cdots\cdots N) \tag{1-5-2}$$

其中 A_{ij} 为方程的系数矩阵，b_i 为常数项列矩阵。

一般将 $A_{i,j}$ 与 b_i 合并成 $A_{i,j}$ 的增广矩阵，则(2)式变成：

$$\sum_{j=1}^{n}A_{i,j}x_j=A_{i,n+1} \qquad (i=1,2,\cdots\cdots N) \tag{1-5-3}$$

解线性方程组的方法可以采用高斯消去法。而高斯消去法又分为一般消去法，列主元消去法和全主元消去法。

(1) 普通高斯消去法：它先用消元法把 $A_{i,j}$ 的增广矩阵化成上三角矩阵，然后用回代法解出 n 个 x 的值。它的方法较简单，但当 $A_{i,j}$ 的对角线元素的绝对值比它同一行的其它元素的绝对值小或等于零时，就会出现超界现象，因此，它只能用在当 A矩阵的主对角元占优势的情况。

(2) 列主元高斯消去法

此方法与普通高斯消去法基本相同，只是进行第K步消元时，对第K列选取绝对值最大元素为主元素，并把主元素所在行(称主行)与第K行的元素互换，然后开始消元过程直至把$A_{i,j}$增广矩阵化为上三角矩阵，最后用回代法求出全部x值。

(3) 全主元高斯消去法

此方法与列主元素消去法基本相同，只是在第K步消元时，把对第K列选取主元的步骤改成对第K，K+1……n行和列选取一个最大元素作为主元素，再把主元所在的行和列与K行和K列互换，然后再进行消元过程，同样用回代法求出全部x值。

用的最多的是列主元消去法和全主元消去法，下面我们把这两种方法的程序和实例列出。两个程序的粗框图是一样的，只是在求主元时和互换时稍有不同。

列主元消去法程序清单

```
1000 REM SOLVE SIMULTANEOUS LINEAR EQUATION BY GAUSSAN METHOD(1)
1010 PRINT"NUMBER OF UNKNOWNS N"
1020 INPUT "N=";N
1030 DIM A(N,N+1)
1035 LPRINT "THE SIMULTANEOUS LINEAR EQUATION IS:"
1040 FOR I=1 TO N
1050 FOR J=1 TO N+1
1060 READ A(I,J)
1062 A$=MID$(STR$(J),2)
1064 IF J=1 LPRINTA(I,J);"X("A$;")";: GOTO 1070
1066 IF J=N+1 LPRINT"="A(I,J): GOTO 1070
1068 IF A(I,j)>=0 LPRINT "+"A(I,j); "X("A$; ")";
     ELSE LPRINT A(I,J); "X("A$; ")";
1070 NEXT J,I
1080 FOR S=1 TO N
1090 FOR T=S TO N
1100 IF A(T,S))(0 THEN GOTO 1140
1110 NEXT T
1120 PRINT"NO UNIQUE SOLUTION"
1130 GOTO 1410
1140 GOSUB 1240
1150 C=1/A(S,S)
1160 GOSUB 1300
1170 FOR T=1 TO N
1180 IF T=S THEN GOTO 1210
1190 C=-A(T,S)
1200 GOSUB 1340
1210 NEXT T
1220 NEXT S
1230 GOTO 1375
1240 FOR J=1 TO N+1
1250 B=A(S,J)
1260 A(S,J)=A(T,J)
1270 A(T,J)=B
1280 NEXT J
1290 RETURN
1300 FOR J=1 TON+1
1310 A(S,J)=C*A(S,J)
1320 NEXT J
1330 RETURN
1340 FOR J=1 TO N+1
1350 A(T,J)=A(T,J)+C*A(S,J)
1360 NEXT J
1370 RETURN
1375 LPRINT "SOLUTION OF EQUATION IS:"
1380 FOR T=1 TO N
1390 X1(T)=A(T,N+1): LPRINT "X(";T;")=" X1(T)
1400 NEXT T
1430 RETURN
1440 DATA 1,1,2,1,4, 2,3,4,2,11, 1,-1,4,5,4, 3,2,1,1,10
```

[注]：程序中句号1430应改为1410

程序框图



全主元消去法程序清单：

```
10 REM  SOLVE SIMULTANEOUS LINEAR EQUATION BY GAUSSAN METHOD
20 INPUT "N=";N
30 DIM A(N,N+1),S(N),M(N)
35 LPRINT "THE SIMULTANEOUS LINEAR EQUATION IS:"
40 FOR I=1 TO N
50 FOR J=1 TO N+1
60 READ A(I,J):A$=MID$(STR$(J),2)
62 IF J=1 LPRINT A(I,J);"X("A$;")";: GOTO70
64 IF J=N+1 LPRINT "="A(I,J): GOTO 70
66 IF A(I,j)>=0 LPRINT"+"A(I,j); "X("A$; ")";
   ELSE LPRINT A(I,J); ")"A$; ")";
70 NEXT J,I
80 FOR I=1 TO N
90 P=I: Q=I: E=A(I,I)
100 FOR J=I TO N
110 FOR K=1 TO N
120 IF ABS(A(J,K))<=ABS(E) THEN 140
130 E=A(J,K): Q=K: P=J
140 NEXT K,J
150 IF ABS(E)>1E-30 GOTO170
160 LPRINT"NO UNIQUE SOLUTION":STOP
170 IF P=I THEN 210
180 FOR K=1 TO N+1
190 S=A(I,K):A(I,K)=A(P,K):A(P,K)=S
200 NEXT K
210 FOR J=1 TO N
220 IF J=I GOTO 280
230 IF A(J,Q)=0 GOTO 280
240 R=A(J,Q)/A(I,Q)
250 FOR K=1 TO N+1
260 A(J,K)=A(J,K)-A(I,K)*R
270 NEXT K
280 NEXT J
290 M(I)=Q
300 NEXT I
310 FOR I=1 TO N
320 Q=M(I): S(Q)=A(I,N+1)/A(I,Q)
330 NEXT I
340 PRINT"THE SOLUTION OF EQUATION IS"
350 LPRINT "THE SOLUTION OF EQUATION IS"
360 FOR Q=1 TO N
370 B$=MID$(STR$(Q),2)
380 PRINT"X("B$;")="S(Q)
390 LPRINT "X("B$;")="S(Q)
400 NEXT Q
410 RETURN
420 DATA 1,1,2,1,4, 2,3,4,2,11, 1,-1,4,5,4, 3,2,1,1,10
```

实例：

设要解的线性代数方程组为：

$$\begin{cases} x_1+x_2+2x_3+x_4=4 \\ 2x_1+3x_2+4x_3+2x_4=11 \\ x_1-x_2+4x_3+5x_4=4 \\ 3x_1+2x_2+x_3+x_4=10 \end{cases}$$

把数据分别从1440句或者420句 DATA语句输入两个程序中，计算结果如下：

THE SIMULTANEOUS LINEAR EQUATION IS:

1x(1)x+1x(2)+2x(3)+1x(4)=4

2x(1)+3x(2)+4x(3)+2x(4)=11

1x(1)-1x(2)+4x(3)+5x(4)=4

3x(1)+2x(2)+1x(3)+1x(4)=10

```
THE SOLUTION OF EQUATION IS
X(1)= 1
X(2)= 3
X(3)=-1
X(4)= 2
```

$x_1=1,\quad x_2=3,\quad x_3=-1,\quad x_4=2$

六、多项式回归，曲线拟合

在一元非线性回归中，若对函数的类型能作出判断时，则可通过一些变换，将它变成线性回归来处理。但若对函数的类型难以作出判断时，则常采用多项式回归来逼近。一般说来，任意曲线都可以近似地用多项式

$$y=b_0+b_1x+b_2x^2+b_3x^2+\cdots\cdots+b_nx^n \qquad (1\text{-}6\text{-}1)$$

来表示。式中，b_0，b_1，b_2，b_3，……，b_n称为回归系数。由最小二乘法原理可知，若要使回归系数为最佳系数，必须使剩差平方和为最小

即 $$Q_{剩}=\sum_{i=1}^{m}[y_i-(b_0+b_1x_{i1}+\cdots\cdots\cdots+b_nx_{ni})]^2=\min \qquad (1\text{-}6\text{-}2)$$

由数学分析极值原理可知，若 b_j 满足方程组

$$\frac{\partial Q_{剩}}{\partial b_j}=0 \qquad (j=0,1,\cdots\cdots\cdots n) \qquad (1\text{-}6\text{-}3)$$

则 b_j 就是最佳的回归系数。

由(1-6-3)式可得一组正规方程组，利用高斯消去法解此正规方程组，就可求得各回归系数。

现在以二次多项式为例说明其回归方法。

设实验中测量值为(x_1,y_1)，(x_2,y_2)，……，(x_n,y_n)

用二次多项式 $$Z=A+Bx+Cx^2 \qquad (1\text{-}6\text{-}4)$$

来近似，其中A、B、C都是回归系数。

由最小二乘法原理可知，剩差平方和为最小，

即 $$Q_{剩}=\sum_{i=1}^{n}(y_i-Z)^2=\min \qquad (1\text{-}6\text{-}5)$$

根据数学分析极值原理：

$$\begin{cases} \dfrac{\partial Q_{剩}}{\partial A}=0 \\ \dfrac{\partial Q_{剩}}{\partial B}=0 \\ \dfrac{\partial Q_{剩}}{\partial C}=0 \end{cases} \qquad (1-6-6)$$

可求得正规方程组为

$$\begin{cases} nA+B\sum_{i=1}^{n}x_i+C\sum_{i=1}^{n}x_i^2=\sum_{i=1}^{n}y_i \\ A\sum_{i=1}^{n}x_i+B\sum_{i=1}^{n}x_i^2+C\sum_{i=1}^{n}x_i^3=\sum_{i=1}^{n}x_i\cdot y_i \\ A\sum_{i=1}^{n}x_i^2+B\sum_{i=1}^{n}x_i^3+C\sum_{i=1}^{n}x_i^4=\sum_{i=1}^{n}x_i^2\cdot y_i \end{cases} \qquad (1-6-7)$$

解正规方程(1-6-7)，求得回归系数 A，B，C

先利用测量值计算

$$X_1=\sum_{i=1}^{n}x_i,\qquad X_2=\sum_{i=1}^{n}x_i^2,\qquad X_3=\sum_{i=1}^{n}x_i^3,$$

$$X_4=\sum_{i=1}^{n}x_i^4,\qquad Y_1=\sum_{i=1}^{n}y_i,\qquad Y_2=\sum_{i=1}^{n}x_i\cdot y_i,$$

$Y_3=\sum_{i=1}^{n}x_n^2\cdot y_i$ 之值

将它们代入(1-6-7)式得：

$$\begin{cases} nA+X_1B+X_2C=Y_1 \\ X_1A+X_2B+X_3C=Y_2 \\ X_2A+X_3B+X_4C=Y_3 \end{cases} \qquad (1-6-8)$$

(1-6-8)式是正规方程，可以调用高斯消去法来求解。

程序说明

程序中变量和数组的物理意义

N 为测量次数

$X(I)$，$Y(I)$分别为第 I 次两个物理量的测量值

$A(I,J)$为正规方程的系数，$I=1,2,\cdots,N$　　$J=1,2,\cdots,N+1$

Q 为剩差平方和